
Introduction
In the course of analyzing data taken at Omega, it becomes necessary to obtain
information on the laser power waveforms from the P510 streak cameras. This memo
describes how to obtain the information and extract waveforms from the HDF files in
which the data is stored using IDL 5 routines.

Information about the P510 traces can be found on Cris Barnes’ web page:

http://www.lanl.gov/orgs/p/p-24/OMEGA/pulseshapes.html

I don’t attempt, in this memo, to explain HDF (Hierarchical Data Format) files. That
information can be obtained from the National Center for Supercomputing Applications
at http://hdf.ncsa.uiuc.edu/.

I do not try to explain how the data is obtained in the first place, how it is analyzed, or
how to do the analysis of the raw data. Instead, I assume that the user is comfortable
using the summary data generated by LLE, and will accept their timing and calibration
numbers as being as good as available.

What I will say is that the P510 streak cameras record the UV pulse shape generated by
the Omega laser facility. Combined with mirror reflection coefficients and transmission
values for various optical components, this data can be used to obtain, for some subset of
the Omega beams, the UV power on target. The analysis of the P510 data is routinely
performed, and the summary data is stored in an HDF file. This memo explains how to
read data from that file using IDL.

Obtaining the P510 summary data from the LLE archives
The P510 data is archived and available from the LLE Image Archives. In order to extract
the data, one first must obtain the necessary HDF file from the archives.

1) Go to the Omega Operations web page at http://omegawww.lle.rochester.edu

2) Log in with a valid username and password

3) Select “Shot Images and Reports”

4) Enter the shot number, click “Admin Summary,” and press the “Query” button.

5) In the left column, under Laser and then P510 Images, click on, for example,
p510_data_17079.hdf.

6) Option-click on the file named, for example, HDF.19990727.172929.1, and rename it
to its original name as shown in the LLE Image Archive (LIA) web page display, i.e.
p510_data_17079.hdf.

Browsing the HDF file

The contents of the HDF file can be examined using an HDF utility such as the free HDF
Browser by Fortner Software, available at:

http://www.fortner.com/docs/product_hdf_b1.html

Opening p510_data_17079.hdf we find the following:

The top seven item are file “attributes” which give information about the shot. Opening
the first three, we see the data stored there. In these cases, we get the shot number, the
clusters on which P510 data was being taken, and the number of beams for which data
was obtained.

The ten lower items are Scientific Data Sets (SDSs), and contain the data which we seek.
Specifically, we are interested in the SDSs which start with “SUMMARY.” These four
contain summaries of the data for the shot.

SUMMARY_DATA_TIME is a vector holding the time values for the points in the other
three summary data sets. The zero-time has been established relative to the fiducial, and
this time vector is common to the rest of the summary data.

SUMMARY_DATA_UV shows the power histories for the active beams for which P510
data is available.

SUMMARY_DATA_AVERAGE is an average power for the active beams.

SUMMARY_DATA_FIDU is the trace of the fiducial curve.

Opening SUMMARY_DATA_UV shows that, for this shot, it is a 20 by 512 array of real
values. This is an array with power waveforms for each active beam in terawatts. This
data is for UV power on  target including transmission losses. These waveforms have
been adjusted to reflect accurate timing relative to the fiducial and to each other.

Expanding the SUMMARY_DATA_UV SDS reveals that it has attributed of its own:

Opening the BEAMS attribute gives a list of active beams in the SDS.

Reading the Power history from the HDF file
The data in the P510 HDF files can be read into IDL using HDF routines included in
IDL 5. The following is an example in which data from “p510_data_17079.hdf’ is read.

1) Start up IDL

2) To open the HDF file, use the routine:
sd_id = hdf_sd_start('p510_data_17079.hdf')

This establishes an id for the file given by sd_id.

3) To read the shot number:
index = hdf_sd_attrfind(sd_id, 'FILE_ATTR_LOG_NUM')

This produces an index to the attribute titled: FILE_ATTR_LOG_NUM in the file
identified by sd_id. Note that the attribute is found by name and is therefore accessible
even if the data structure changes.

hdf_sd_attrinfo, sd_id, index, DATA=shotnumber

This assigns the value stored in the attribute associated with index in the file associated
with sd_id to the variable shotnumber.

4) To read the time array from the SDS SUMMARY_DATA_TIME, we use the
routine:

index = hdf_sd_nametoindex(sd_id, 'SUMMARY_DATA_TIME')

This creates an index not to an attribute, as above, but to an SDS. We then select the SDS
pointed to by index:

sds_id = hdf_sd_select(sd_id, index)

To get the data in the SDS, we then use:

hdf_sd_getdata, sds_id, summary_data_time

This creates an IDL array with the time values for the points in the other SDSs.

5) To read the UV on target waveforms, we use the same steps as for the
time:

index = hdf_sd_nametoindex(sd_id, 'SUMMARY_DATA_UV')

sds_id = hdf_sd_select(sd_id, index)

hdf_sd_getdata, sds_id, summary_data_uv

This creates an IDL n x 512 array, where each column contains the power history of the
active beams with the time for each step given by the summary time curve read in step 4.

In addition to the power curves, we need to know which beam’s waveform is given in
each column. This is done with the steps below:

index2 = hdf_sd_attrfind(sds_id,'BEAMS')

Note that to establish the index to the attribute BEAMS, we use the index sds_id,
which references the SDS of which it is an attribute, rather than the index to the file.

hdf_sd_attrinfo, sds_id, index2, DATA=beams

The list of beams is then placed in the array beams.

Average power and the fiducial waveform can be obtained by a similar set of commands.
Each of these SDSs has its own set of attributes as well.

I have written an IDL routine that reads a P510 HDF file and converts the summary data
into an Igor text file for use with the data analysis program Igor by WaveMetrics (which I
use for analyzing data from the LANL Bang Time Detector on Omega). When read into
Igor, it generates a plot of the waves and the fiducial vs. time. However, once the data is
read into IDL with the above routines, the user is able to use the data in any manner
needed.

Output from a shot with 20 beams recorded on the P510 streak cameras using PS26N01.

Output from a shot with 4 backlighter beams recorded on the P510 streak cameras using
pulse shapeRM2001.

Appendix: Text of my IDL routine p510igor.pro
PRO p510igor, filename

; Open the file

sd_id=hdf_sd_start(filename)

; Get the shot number

index=hdf_sd_attrfind(sd_id,'FILE_ATTR_LOG_NUM')
hdf_sd_attrinfo,sd_id,index, DATA=shotnumber

; Get the Time array

index = hdf_sd_nametoindex(sd_id,'SUMMARY_DATA_TIME')
sds_id=hdf_sd_select(sd_id,index)
hdf_sd_getdata,sds_id,summary_data_time
s = size (summary_data_time)
npts = s[1]

; Get the UV traces

index = hdf_sd_nametoindex(sd_id,'SUMMARY_DATA_UV')
sds_id=hdf_sd_select(sd_id,index)
hdf_sd_getdata,sds_id,summary_data_uv

; Get the beam numbers included in the Summary UV data

index2 = hdf_sd_attrfind(sds_id, 'BEAMS')
hdf_sd_attrinfo, sds_id, index2, DATA=beams
s = size(beams)
nbeams = s[1]
print, 'Number of beams: ', nbeams
print, 'Active beams: ',beams

index = hdf_sd_nametoindex(sd_id,'SUMMARY_DATA_AVERAGE')
sds_id=hdf_sd_select(sd_id,index)
hdf_sd_getdata,sds_id,summary_data_average

index = hdf_sd_nametoindex(sd_id,'SUMMARY_DATA_FIDU')
sds_id=hdf_sd_select(sd_id,index)
hdf_sd_getdata,sds_id,summary_data_fidu

; Get the Fiducial Peak Locations

index2 = hdf_sd_attrfind(sds_id, 'PKS')
hdf_sd_attrinfo, sds_id, index2, DATA=pks
s = size(pks)
npeaks = s[1]

; Write contents to a file

openw, 1, 'P510_data.asc'

printf,1,FORMAT='("IGOR")'



Reading P510 Data Using IDL 5 Routines LA-UR-99-6136

14

printf,1, FORMAT='($,"WAVES/O Timing   Fidu     ")'
for j=0, nbeams-1 DO printf,1,FORMAT='($,"B",i02,"      ")',beams[j]
printf,1, FORMAT='(" ")'

printf,1, FORMAT='("BEGIN")'

for i=0, npts-1 DO BEGIN
printf,1, FORMAT='($,f8.4," ")', summary_data_time[i]/1000.
printf,1, FORMAT='($,f8.5," " )', summary_data_fidu[i]
for j=0,nbeams-1 DO printf,1, FORMAT='($,f8.5," ")',

summary_data_uv[j,i]
printf,1, FORMAT='(" ")'

ENDFOR

printf,1, FORMAT='("END")'

printf,1, "WAVES/O FiduTimes"
printf,1, "BEGIN"
FOR i = 0, npeaks-1 DO printf,1, FORMAT='(f8.4)', pks[i]/1000.
printf,1, FORMAT='("END")'

printf,1, FORMAT="('X Variable/G P510ShotNumber = ', I6)", shotnumber

printf,1, FORMAT="('X Display B',i02, ' vs Timing')", beams[0]

if nbeams gt 1 THEN for j = 1, nbeams-1 DO printf, 1, FORMAT="('X
Append B',i02, ' vs Timing')", beams[j]

printf,1, "X Label left ""Beam Power (TW)"""
printf,1, "X Label bottom ""Time (ns)"""

printf,1, "X AppendToGraph/R Fidu vs Timing"
printf,1, "X ModifyGraph rgb(Fidu)=(0,0,65535)"

printf,1, "X ModifyGraph fStyle=1, axThick=2, fSize=14, width=750,
height=300"
printf,1, "X ModifyGraph width=0, height=0"

printf,1, "X SetAxis left -.02,.4"
printf,1, "X SetAxis right -.2, 4."
printf,1, "X SetAxis bottom -2., 4."

printf,1, FORMAT="('X Textbox/N=text0/F=2/S=3/A=LT ""\\f01\\Z14Shot ',
i5,'""')", shotnumber

printf,1, "X ShowInfo"

close,1

cmd = 'mv P510_data.asc '+ 'p510_data_' +
string(FORMAT='(i05)',shotnumber) + '.igor'
print, cmd
SPAWN, cmd
return
END


